

Creating Customized Word Tables with Table of Contents using SAS ODS RTF for Clinical Studies

Sue X. Zhang, Abbott Nutrition Research and Development, Abbott Laboratories,
Columbus, Ohio

ABSTRACT

SAS® ODS RTF destination allows the output tables to be directly copied into Word documents by the users. It creates the linked table of contents (TOC) pages that take you directly to the desired output with just a mouse click. The TEMPLATE procedure can customize SAS output reports by having control over the margins, font type, font size and justifications etc.

This paper demonstrates the step by step procedures on how to apply the ODS RTF CONTENTS option and PROC TEMPLATE to generate informative TOC and customized reports with specified formats. It also describes the methods for generating publication quality tables using PROC TABULATE and PROC REPORT in addition to the ODS RTF CONTENTS option and PROC TEMPLATE. Some issues with these procedures will be discussed. The data used in this paper is either directly obtained or modified from SASHELP.CLASS for demonstration purposes only.

PC SAS version 9.13 and Microsoft® Office Word 2003 on Windows XP Professional version 2002 were used in this paper.

INTRODUCTION

For years before the introduction of ODS, the clinical SAS programmers generated reports in the SAS default ASCII text output (i.e. the SAS LST file). Although the ASCII file is converted to a Word file using Microsoft Word, the tables could not be easily copied into a clinical study report (CSR). The tables still needed to be retyped to fit the CSR format with specific font type and other requirements. This is time consuming and easily generates typing errors.

ODS RTF generates Word files directly and it provides a great advantage over the traditional ASCII file. The tables from the RTF file can be formatted into certain font type and size consistent with that of the CSR that the clinical scientists produced. The tables can be customized and directly copied into the CSR without retyping.

This paper will demonstrate:

1. How to generate customized TOC using the ODS RTF CONTENTS option and PROC TEMPLATE for PRINT, REPORT and TABULATE SAS procedures.
2. How to customize the report with certain formats using the TEMPLATE SAS procedure.
3. How to create publication quality summary tables using the TABULATE and REPORT SAS procedures in addition to the methods mentioned above.
4. Issues with TOC, title and footnote and how they were resolved.


METHODS

DEMO 1: GENERATING TOC WITH SAS PROCEDURE NAMES USING DEFAULT TEMPLATE

TOC is created with the ODS RTF CONTENTS option. The following output demonstrates TOC using PROC PRINT, PROC REPORT and PROC TABULATE. The code is not shown here but is described in Demo2 with the exception of the bold areas using ODS PROCLABEL and CONTENTS='.....'.

The RTF output has a Table of Contents page. With the RTF document open and your cursor in the body of the TOC page, right click the mouse and select Update Field (SAS support web site, <http://support.sas.com/rnd/base/ods/odsrtf/rtf901.html>):

FIGURE 1: DEMO1: GENERATING TOC WITH SAS PROCEDURE NAMES



By default, The TOC only presents the procedure names here.

<i>Table of Contents</i>	
Print.....	2
Data Set SASHELP.CLASS	2
Report	3
Detailed and/or summarized report.....	3
Tabulate	4
Cross-tabular summary report.....	4
Table 1	4

DEMO 2: GENERATING TOC WITH INFORMATIVE TABLE TITLES

By using ODS PROCLABEL and CONTENTS=' '(Ensor et al. 2006), the TOC can provide more informative headings about the tables generated. The macro DatePage is developed to show the project name, report descriptions, author, date and page number in the Word Header place by using TITLE statements. See the code below:

```
/* Create a macro to add the titles on every page (except the cover page) with */
/* Project number, report descriptions, author, date and page number */
%macro datepage;
```

```

/* Leftmargin=3.5in specifies the left margin of the title */
title1 j=1 "^S={leftmargin=3.5in}Abbott Nutrition (Project xxxx)";
title2 j=1 "^S={leftmargin=3.5in}Statistical Report: Appendix 1";
/* &Sysdate9 displays date in the date9. format */
title3 j=1 "^S={leftmargin=3.5in}Created By xxxx xxx on &sysdate9.";

/* Add page number (Page X of Y) (Zender 2007) */
title4 j=1 "^S={leftmargin=3.5in}Page ^{thispage} of ^{lastpage}";
title5 " ";
%mend datepage;

/* Orientation= specifies the page orientation */
options nodate nonumber byline orientation=portrait;

/* Add in-line formatting (Zender 2007) */
ods escapechar="^";

ods listing close;

/* The CONTENTS option generates TOC */
ODS rtf file="c:\ODSdemo\output\Demo2.rtf" startpage=yes contents;

/* ODS PROCLABEL '.....' generates the first level of the title from TOC */
ods proclabel 'Demo 2: Table 1 - Generate Toc with Informative Table Titles and
Use Default Template - PROC PRINT';

%DatePage;
title6 'Demo 2: Table 1 - Generate Toc with Informative Table Titles and Use
Default Template - PROC PRINT';

/* CONTENTS ="....." generates the second level of the title from TOC. It can be */
/* omitted by specifying CONTENTS=" " . */
proc print data=sashelp.class noobs label contents ='Subjid - Gender print';
var name /style={font_weight=bold cellwidth=3cm just=1};
var sex/style={cellwidth=3cm just=1};
run;

ods proclabel 'Demo 2: Table 2 - Generate Toc with Informative Table Titles and
Use Default Template - PROC REPORT';

%DatePage;
title6 'Demo 2: Table 2 - Generate Toc with Informative Table Titles and Use
Default Template - PROC REPORT';

PROC REPORT data=sashelp.class nowd headskip missing
contents ='Subjid - Gender Report';
column name sex;
define name /order order=internal style(column)={font_weight=bold
cellwidth=3cm just=left};
define sex /style(column)={cellwidth=3cm just=1};
run;

ods proclabel 'Demo 2: Table 3 - Generate Toc with Informative Table Titles and
Use Default Template - PROC TABULATE';

%DatePage;
title6 'Demo 2: Table 3 - Generate Toc with Informative Table Titles and Use
Default Template - PROC TABULATE';

PROC TABULATE data=sashelp.class missing contents ='Subjid - Evaluable group';
CLASS sex;

/* CONTENTS ="....." generates the third level of the title from TOC. It can */
/* be omitted by specifying CONTENTS =" " . */
TABLE sex all='Total'/ contents ='Tabulate 3rd label';
run;

```

```
ods _all_ close;
ods listing;
```

FIGURE 2: DEMO2: TABLE OF CONTENTS WITH INFORMATIVE LABELS FOR TABLES

<i>Table of Contents</i>	
Demo 2: Table 1 - Generate Toc with Informative Table Titles and Use Default Template - PROC PRINT	2
Subjid - Gender print	2
Demo 2: Table 2 - Generate Toc with Informative Table Titles and Use Default Template - PROC REPORT	3
Subjid - Gender Report	3
Demo 2: Table 3 -Generate Toc with Informative Table Titles and Use Default Template - PROC TABULATE	4
Subjid - Evaluable group	4
Tabulate 3rd label	4

From Word 2003, by pressing Ctrl and click on the title (From Word 2000, just click the title), it will take you directly to the desired table output.

FIGURE 3: DEMO2: TITLES AND PARTIAL TABLE OUTPUT

The titles with project name, report descriptions, author, date and page numbers repeat on each page in Word Header place from page 2

Abbott Nutrition (Project xxxx)
Statistical Report: Appendix 1
Created By xxxx xxx on 11NOV2009
Page 2 of 4

Demo 2: Table 1 - Generate Toc with Informative Table Titles and Use Default Template - PROC PRINT

The table title appears in Word Header place using TITI F6 statement

Name	Sex
Alfred	M
Alice	F
Barbara	F
Carol	F
Henry	M

By default, the title font is Bold Italic Times New Roman with size of 13. The column header font is Times New Roman with size of 11. The font of the text in the table is Times New Roman with size of 10.

DEMO 3: GENERATING INFORMATIVE TOC AND CUSTOMIZED TABLE STYLES

The TEMPLATE procedure below provides control over the aspects of all the tables generated from the same file unless otherwise specified from the SAS procedures using the Style options. The customized style template odsdemo is created. (Ensor et al. 2006 and with the help from Michele Ensor and the SAS support team).

```
proc template;

  /* Specify the font type (Arial) and size (10pt) */
  %let fsize=10pt;
  %let fntype=arial;

  /* Define the name of the customized style template: odsdemo */
  define style odsdemo;

    /* Modify from SAS pre-defined style: styles.rtf */
    parent=styles.rtf;

    style contenttitle from indextitle /

    /* Add ABBOTT NUTRITION company logo */
    preimage="c:\ODSdemo\ABBOTTNutrition_LOGO.JPG"

    /* Specify the text appearing at the beginning of the report below the */
    /* company's Logo. Where ^n directs the text to a new line. */
    pretext="^nDemo 3: Project # - XXXXXX^n
              Statistical Report^n
              Appendix 1 - XXXXXX^n
              Created by xxxx xxx on &sysdate9.^n ^n
              TABLE OF CONTENTS^n ^n "

    /* Specify the font type (Arial bold) and size (12pt) of the text above */
    font = ("&fntype",12pt) font_weight=bold foreground=black ;

    style contents / background=white font = ("&fntype",&fsize);

    /* Control font type (Arial) and size (10 pt) for text specified with ODS */
    /* TEXT of the table. */
    style usertext / font = ("&fntype",&fsize) just=c;

    style systemtitle / font = ("&fntype",&fsize) font_weight=bold
                          foreground=black;
    style systemfooter / foreground=black font = ("&fntype",&fsize );

    style data / background=white font = ("&fntype",&fsize );

    style cell / background=white font = ("&fntype",&fsize );

    style table / background=white font = ("&fntype",&fsize )
      /* Specify table rules and borders */
      rules=all frame= hside /* for Demo 3 */

      /* rules=group frame= hside /* for Demo 4 */

      cellspacing=0 cellpadding=3pt borderwidth=1pt;

    style byline from titlesandfooters "controls byline text." /
      font=("&fntype",&fsize );

    style batch from batch / font = ("&fntype",&fsize );

    style header / background=white foreground=black
      font_weight=bold font=("&fntype", &fsize);

    style rowheader / background=white font_weight=bold font=("&fntype",
```

```

&fsize);

style body from document / font = ("&fntype",&fsize )

/* Specify the margin of the RTF file */
topmargin = 1.0in
bottommargin = 1.0in
leftmargin = 1.0in
rightmargin = 1.0in;
end;
run;

options nodate nonumber byline orientation=portrait;
ods escapechar="^";


ods listing close;
ods rtf file="c:\odsdemo\output\demo3.rtf" startpage=yes contents
      style=odsdemo;

..... /* SAS procedures from Demo 2 */

ods _all_ close;
ods listing;

```

FIGURE 4: DEMO3: GENERATING INFORMATIVE TOC AND CUSTOMIZED TABLE STYLES

	
Demo 3: Project # - XXXXXX Statistical Report Appendix 1 - XXXXXX Created by xxxx xxx on 12NOV2009	
TABLE OF CONTENTS	
Demo 3: Generate Informative TOC and Customized Table Styles - PROC PRINT	2
Subjid - Gender print	2
Demo 3: Table 2 - Generate Informative TOC and Customized Table Styles - PROC REPORT	3
Subjid - Gender Report	3
Demo 3: Table 3 - Generate Informative TOC and Customized Table Styles - PROC TABULATE	4
Subjid - Evaluable group	4
Tabulate 3rd label.....	4

As specified from PROC TEMPLATE, the text on and above TABLE OF CONTENTS area is bold with Arial size 12, the title and row header of the table is bold with Arial size 10 and the body of the table is regular Arial size 10. Table 1 appears on page 2 of 4. Below is the partial table output on page 2.

FIGURE 5: PARTIAL OUTPUT OF TABLE 1 USING PROC PRINT

The table title is in Word Header place

Abbott Nutrition (Project xxxx)
Statistical Report: Appendix 1
Created By xxxx xxx on 11NOV2009
Page 2 of 4

Demo 3: Table 1 - Generate Informative TOC and Customized Table Styles - PROC PRINT

Name	Sex
Alfred	M
Alice	F
Barbara	F
Carol	F

DEMO 4: GENERATING CUSTOMIZED SUMMARY TABLES

Within our company, the clinical scientists who use the summary tables produced by the programmers want the tables in a certain format so that they can directly use the selected tables in their CSR as well as publications without retyping. For example, the format for a CSR is Arial 11 for text and Arial 10 or 11 for tables. Categorical variables need to be presented as n(%). Continuous variables should be in the form of MEAN±SEM for the presentation of the mean and standard error etc.. PROC TABULATE is used to generate a summary data set. This data set is then manipulated into the desired format. PROC REPORT is then applied to generate the tables with certain table format requirements.

Another requirement for the tables is that the table specific titles and footers (sometimes the statistical analysis results are incorporated with the summary tables as footers) should appear in the body of the tables instead of in page headers or page footers. Thus the footers can be copied with the table together at the same time without being copied twice by going to Word Header or Footer place. Different methods were tested and we ended up using the TITLE statements for project number, report descriptions, author, date and page numbers that repeat on every page starting from page 2. These titles appear in the Word Header place and are not needed for the CSR. STYLE(REPORT)={PRETEXT="....."} was used to generate table specific titles which appear in the body of the table and ODS RTF TEXT="....." was used to get footers in the body of the table. (SAS support: Sample 36288). See the code below:

```
proc format;
  value trt 1='Placebo' 2='Test Drug' 99999='Total';
  value sex .='Missing' 1='Male' 2='Female' 99999='Male/Female Combined';
  value $all " " "=";
run;

data class_1;
  set sashelp.class;

  /* Create numeric variable for sex */
  if sex='M' then sexnum=1;
  else if sex='F' then sexnum=2;

  if _n_ in(1 3 5 7 9 11 13 15 17 19) then trtgrp=1;
  else trtgrp=2;
run;

ods listing close;

/* Exclude the table from the ODS destination. */
ods exclude table;

/* Use PROC TABULATE to output summary data set for Gender by Treatment */
proc tabulate data=class_1 out=sex(drop=_page_ _table_) exclusive;
  class sexnum /order=data;
```

```

class trtgrp/order=data missing;
table (sexnum all='total'), (trtgrp all='total')*(n pctn<sexnum all>='% of
column total') all='total'/misstext='0' ;
quit;

data sex;
set sex;
length varcat value $250;
if _type_=00 then do;
sexnum=99999; trtgrp=99999;
end;
else if _type_=10 then do;
trtgrp=99999;
end;
else if _type_=01 then do;
sexnum=99999;
end;
if trtgrp = 99999 then pct=pctn_00;
else pct=pctn_01;
format sexnum sex. trtgrp trt.;
varcat=put (sexnum,sex.);
value=trim(left(put (n,6.0)))||'('||trim(left(put (pct, 6.0)))||')';
sectno = 1;
subsecno=trtgrp;
keep sectno subsecno sexnum trtgrp varcat value;
run;

ods exclude table;

/* Use PROC TABULATE to output summary data set for Weight by Treatment group */
proc tabulate data=class_1 out=weight(drop=_page_ _table_) exclusive;
class trtgrp/missing;
var weight;
tables (trtgrp all), (weight='')*(mean*f=8.0 median*f=8.0 std*f=8.0
stderr*f=8.0 min*f=8.0 max*f=8.0 n*f=10.0 nmiss*f=10.0
sum*f=10.0);
quit;

data weight;
set weight;
length varcat value $250;
if _type_=0 then trtgrp=99999;
format trtgrp trt.;
sectno = 2;
subsecno=1;
subsecno=subsecno+1;

/* The special character '\pm' is copied from WORD symbol. Parsons (2008) */
/* has discussed how to create special character in the RTF file from SAS. */
varcat = "Mean  $\pm$  SEM";

if weight_mean ne . then value=trim(left(put (weight_mean,8.0)))||'  $\pm$  '
||trim(left(put (weight_stderr,8.0)));
output;

subsecno=subsecno+1;
varcat = "Median";
value=trim(left(put (weight_median,8.0)));
output;

subsecno=subsecno+1;
varcat = "Std Dev";
value=trim(left(put (weight_std,8.0)));
output;

```



```

subsecno=subsecno+1;
varcat = "Min, Max";
if weight_min ne . or weight_max ne . then
value=trim(left(put(weight_min,8.0))||', '||trim(left(put(weight_max,8.0))));
output;
subsecno=subsecno+1;
varcat = "n";
value=trim(left(put(weight_n,6.0)));
output;
drop weight_mean weight_median weight_std weight_stderr
      weight_min weight_max weight_n weight_nmiss weight_sum _type_;
run;

/* Combine the two data sets */
data bytrtgrp;
  set sex weight;
run;

proc sort data=bytrtgrp;
  by sectno trtgrp subsecno;
run;
data temp;
  set bytrtgrp;
  by sectno trtgrp subsecno;

  /* Create variable for first column of the summary table */
  length fstcolumn $200.; if first.sectno and sectno=1 then
fstcolumn='^S={font_weight=bold}Gender, n(%)^S={}';
  else if first.sectno and sectno=2 then
fstcolumn='^S={font_weight=bold}Weight (lb)^S={}';

  if first.sectno;
  keep sectno trtgrp subsecno fstcolumn;
run;

data bytrtgrp;
  set temp bytrtgrp(in=bb);
  by sectno trtgrp subsecno;
  length fstcolumn $200.;
  if bb then  fstcolumn='          '||trim(left(varcat));

  /* Drop the treatment total if not needed */
  if trtgrp=99999 then delete;

  drop sexnum;
  format trtgrp trt.;
  label trtgrp='Treatment Group';
run;

options nodate nonumber byline orientation=portrait;
ods escapechar="^";

ods listing close;
ods rtf file="c:\odsdemo\output\demo4.rtf" startpage=yes contents
      style=odsdemo;

ods proclabel 'Demo 4 - Customized Summary Tables: Gender and Weight';

%datepage;

/* Pretext='.....' adds the titles in the body of the table. \line\ starts a */
/* new line of the titles */
proc report data=bytrtgrp nowd headskip missing
  contents="by treatment group"
  style(report)={font_weight=bold

```

```

pretext="Demo 4 - Customized Summary Tables: Gender and Weight\line\
      By Treatment Group";

column sectno fstcolumn trtgrp, value(dummy);
define sectno/noprint group order=data;
define fstcolumn/group ' ' order=data style(header)={asis=on just=left
      vjust=middle}
style(column)={asis=on cellwidth=5cm just=1};
define trtgrp/across order=data;
define value/format=$all. ' ' style(column)={cellwidth=3cm just=c};
define dummy / noprint;

run;

/* Add footers in the body of the table */
ods rtf text="^S={outputwidth=65% just=1}Footer 1 in the body of the table";
ods rtf text="^S={outputwidth=65% just=1}Footer 2 in the body of the table";

ods _all_ close;
ods listing;

```

The output from the code above:

FIGURE 6: DEMO 4: CUSTOMIZED SUMMARY TABLES

Titles are in Word Header place

Abbott Nutrition (Project xxxx)
Statistical Report: Appendix 1
Created By xxxx xxx on 11NOV2009
Page 2 of 2

Titles are in the body of the table

**Demo 4 - Customized Summary Tables: Gender and Weight
By Treatment Group**

	Treatment Group	
	Placebo	Test Drug
Gender, n(%)		
Male	6(60)	4(44)
Female	4(40)	5(56)
Male/Female Combined	10(100)	9(100)
Weight (lb)		
Mean ± SEM	100 ± 9	100 ± 5
Median	100	100
Std Dev	29	16
Min, Max	51, 150	83, 128
n	10	9

Footers are in the body of the table

Footer 1 in the body of the table
 Footer 2 in the body of the table

DISCUSSIONS

An issue with the TOC is that the TOC generated in the RTF file from PC SAS version 9.13 may need to be manually set to the desired formats. If using `OPTIONS ORIENTATION=landscape`, the first page of the TOC needs to be set manually to landscape orientation from Word File → Page Setup. In addition, all the TOC pages may need to be set manually to the desired font type, font size and font style (Arial 10pt regular in our report).

Another issue with the TOC is that when copying the TOC from the RTF file to another Word file, the TOC page numbers cannot be viewed from Word File→ Print Preview or printed out. It alerts "Error! Bookmark not defined". Since the whole summary table file is used as an independent Appendix for our clinical studies, there is no need to insert the whole file including the TOC into another Word document. Only certain summary tables are selected and copied into the CSR. Parsons (2008) has discussed more techniques on creating TOC in her paper.

To get the titles repeat on each page with the same project name, report descriptions, author, date and page numbers to, the TITLE statements were used. The titles appear in the Word Header place. To get the same footers to repeat on each page, the FOOTNOTE statements were used. The footers are in the Word Footer place.

To get the table specific titles and footers in the body of the table, the BODYTITLE option was tested. The page numbers cannot be presented on every page of the report. The BODYTITLE option places the title only on the first page if the table spans more than one page. Also, the footnote only appears on the last page of the table. Smoak (2004) has discussed the limitations using BODYTITLE.

To place titles in the body of the table, `ODS RTF PREPAGE = '.....'` in front of the SAS procedure was tested. The problem is if a table has multiple pages, the column headers only appear once.

With the help from the SAS support team, we use `STYLE={PRETEXT='.....'}` for titles that appear in the body of the table. The titles only appear once from the table and work for us since in most cases, each of our summary tables only takes one page. To repeat the titles at the top of every table on each page using PROC REPORT, see SAS support sample 36288 at <http://support.sas.com/kb/36/288.html>

Feder (2004) has used `ODS RTF TEXT='.....'` for titles and footers in the body of the table. For titles, we found that `ODS RTF TEXT='.....'` does not work for our report with multiple tables. Sometimes statistical analysis results need to be incorporated into our summary tables as footers and need to be copied with the table to the CSR. `ODS RTF TEXT='.....'` works for footers in the body of the table. If a table expands across multiple pages, the footer only appears at the bottom of the table. To specify the correct justification, use `ODS RTF TEXT="^S={outputwidth=xx% just=l}....."`; In the customized table as shown in Demo 4, `outputwidth=65%` is specified so that the width of the footer cell is the same as the width of the table. If `outputwidth=100%` was specified, when coping the RTF file to another Word file such as the CSR, the footer cell expands the whole page. The table itself is forced left justified instead of center justified from the new Word table.

It is worth mentioning that `STYLE={POSTEXT='.....'}` also works for the footer in the body of the table. More details about using `PREPAGE=`, `PRETEXT=`, `POSTTEXT=` and `TEXT=` can be found from SAS support web site: <http://support.sas.com/rnd/base/ods/odsrtf/rtf901.html>

CONCLUSIONS

By using the `ODS RTF CONTENTS` option and `PROC TEMPLATE`, the programmers can generate a customized clinical report output with Table of Contents. The TOC pages provide the user with information on what kinds of tables are produced. The user can view the specific tables by clicking the title of the table from the TOC without scrolling down pages and pages of the tables to find the right table. It is very convenient to use. The customized tables generated can be directly copied into the CSR for the clinical studies and publications with little editing or with no editing.

REFERENCES

- Carey G. Smoak, 2004. "Creating Word Tables using PROC REPORT and ODS RTF." PharmaSUG Proceedings. San Diego, California.
- Cynthia L. Zender. 2007. "Funny ^Stuff~ in My Code: Using ODS ESCAPECHAR." SAS Global Forum Proceedings. Orlando, Florida.

- Lauren Haworth. 2004. "SAS® with Style: Creating your own ODS Style Template for RTF Output." PharmaSUG Proceedings. San Diego, California.
- Lori S. Parsons. 2007. "Enhancing RTF Output with RTF Control Words and In-Line Formatting." SAS Global Forum Proceedings, Orlando, Florida.
- Lori S. Parsons. 2008. "Creating a Custom Table of Contents in RTF Documents." SAS Global Forum Proceedings. San Antonio, Texas.
- Michele Ensor, Cynthia Zender. 2006. Advanced Output Delivery System (ODS) Topics, Course Notes for Live Web training on 5/22/06 – 5/25/06. Cary, NC: SAS® Institute Inc.
- SAS support. Base SAS, Preproduction RTF Features in SAS 9.1.
<http://support.sas.com/rnd/base/ods/odsrtf/rtf901.html>. Accessed March 2008.
- SAS support. Sample 36288: Repeating text on an RTF page when the BODYTITLE option is in effect.
<http://support.sas.com/kb/36/288.html>. Accessed August 2009.
- Steven Feder. 2004. "Perfecting Report Output to RTF." SUGI 29 Proceedings. Montréal, Canada.

ACKNOWLEDGEMENTS

I would like to thank my manager Wendy Jones for her consistently encouraging and supporting, for her carefully reviewing the paper with valuable comments and suggestions. I would like to acknowledge the professional support and advice from the SAS support team. I appreciate the help from Michele Ensor from SAS during and after the Live Web training on Advanced Output Delivery System (ODS) Topics. Thanks go to everyone in my company for their help with the paper submission process.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Please contact the authors at:

Sue X. Zhang
Abbott Nutrition Research and Development
Dept. 105400, Bldg. RP3-2
3300 Stelzer Road
Columbus, OH 43219
Work Phone: 614-624-6531
Email: Sue.zhang@abbott.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Microsoft® Office is a registered trademark of the Microsoft Corporation.

Other brand and product names are registered trademarks or trademarks of their respective companies.