

# Improving Health Care Quality with the RAREEVENTS Procedure

Bucky Ransdell, SAS Institute Inc.

## ABSTRACT

Statistical quality improvement is based on understanding process variation, which falls into two categories: variation that is natural and inherent to a process, and unusual variation due to specific causes that can be addressed. If you can distinguish between natural and unusual variation, you can take action to fix a broken process and avoid disrupting a stable process. A control chart is a tool that enables you to distinguish between the two types of variation. In many health care activities, carefully designed processes are in place to reduce variation and limit adverse events. The types of traditional control charts that are designed to monitor defect counts are not applicable to monitoring these rare events, because these charts tend to be overly sensitive, signaling unusual variation each time an event occurs. In contrast, specialized rare events charts are well suited to monitoring low-probability events. These charts have gained acceptance in health care quality improvement applications because of their ease of use and their suitability for processes that have low defect rates. The RAREEVENTS procedure, which is new in SAS/QC<sup>®</sup> 14.1, produces rare events charts. This paper presents an overview of PROC RAREEVENTS and illustrates how you can use rare events charts to improve health care quality.

## INTRODUCTION

The influential management consultant W. Edwards Deming advocated an approach to quality improvement that is based on three principles:

1. All work occurs in a system of interconnected processes.
2. Variation is present in all processes.
3. Understanding and reducing variation are critical to success.

Deming built upon the earlier work of Walter Shewhart, who is considered the originator of statistical quality control. Shewhart recognized two distinct types of process variation. Chance cause (or common cause) variation is naturally present in a process. Assignable cause (or special cause) variation is sporadic and occurs for specific, identifiable reasons. Identifying and eliminating assignable cause variation result in a stable, predictable process. However, attempting to eliminate the inherent chance cause variation from a stable process is counterproductive. Shewhart developed the control chart (or Shewhart chart), a graphical and analytical tool that distinguishes between the natural variation in a process and unusual variation due to assignable causes.

Although statistical process control was developed for monitoring manufacturing processes, its ideas and tools are applicable to any industry. In particular, control charts are widely used in health care quality improvement (Woodall 2006). Many different varieties of control chart are used to analyze different kinds of process data, but they fall into two broad categories. A chart for variables is used for a quality characteristic that is measured on a continuous scale, such as waiting times for emergency room patients. A chart for attributes is traditionally used for a quality characteristic that is measured by counting the number of defects in an item or the number of defective items in a sample. In health care quality improvement, attribute charts are used to analyze counts of adverse events, such as patient falls or medication errors.

## RARE EVENTS CHARTS

Many health care processes are high-quality processes, meaning that adverse events occur infrequently, with a low probability. As explained by Kaminsky et al. (1992), Benneyan (2001), and others, traditional attribute charts do not always perform well when used to monitor these rare events. A specialized control chart that is designed for monitoring rare events is called a rare events chart. Rare events charts have gained acceptance in health care

quality improvement applications because of their ease of use and suitability to processes that have low defect rates (Benneyan 1999). The RAREEVENTS procedure, which is available in SAS/QC 14.1, produces rare events charts.

The ideal data for a rare events chart are counts of the opportunities for an event of interest to happen that fall between events. An example is the number of surgeries performed between instances of postsurgical infection. However, these kinds of data are not always available. For example, if you want to monitor patient falls in a hospital, there is unlikely to be an explicit count of opportunities for a fall. In that case, you can use a measure such as the number of days or, preferably, the number of patient-days between falls.


An important assumption for a rare events chart is that the events are independent: the occurrence of one event does not affect the probability that another will occur. Also, the probability of an event must be approximately constant over time. You should not use rare events charts to monitor clusters of events, such as cases of a contagious disease, which violate this assumption. For a thorough discussion of different control charts that are applicable to health care quality improvement, see Woodall (2006).

## MONITORING ACCIDENTAL NEEDLE STICKS

Accidental needle sticks present a substantial risk for health care workers, because they can transmit diseases such as hepatitis and HIV. Health care facilities in developed countries take many precautions to minimize the incidence of accidental sticks, such as providing appropriate disposal containers, using sheathed or retractable needles, and training workers in safe procedures (Wilburn and Eijkemans 2004).

In the following example, a hypothetical hospital monitors needle stick accidents to determine whether its process for handling needles is stable and consistent. It records the number of needle sticks that occur each week. [Figure 1](#) shows needle stick data over 60 weeks.

**Figure 1** Weekly Counts of Accidental Needle Sticks




In this case needle sticks are rare. Almost all the weekly counts are 0, and no week has more than 1. Because these are count data, a traditional control chart approach would employ a *c* chart, which is used to monitor counts of defects. The following statements use the SHEWHART procedure to produce a *c* chart of needle sticks per week:

```
proc shewhart data=NeedleSticks;
  cchart Sticks * Week /
    markers nolegend totpanels=1
    vaxis=0 1 2 cout
    odstitle='c Chart for Needle Sticks';
run;
```

A lower control limit (LCL) and upper control limit are computed from the **Sticks** data. Values greater than the UCL or less than the LCL signal unusual variation. The *c* chart is shown in [Figure 2](#).

**Figure 2** Traditional  $c$  Chart of Needle Sticks per Week



Because 55 of the 60 weekly counts are 0, the mean count is less than 0.1 and the UCL is 0.9. With a UCL that is less than 1, each individual needle stick signals unusual variation. Therefore, this chart is of no use in distinguishing between chance cause variation and assignable cause variation. It is too sensitive.

To address this problem, you could increase the subgroup sample size. For example, you could count the number of needle sticks in two-week periods. The way that measurements are organized into subgroups is critical to the interpretation of a control chart. You should not arbitrarily change how subgroups are defined, although in this case combining counts from consecutive weeks into two-week counts would probably not cause a problem. A drawback of this approach is that data are available for analysis only every two weeks, so you might not detect a change in the process quickly. A better alternative is to use a rare events chart.

Figure 3 shows a listing of the data set `WeeksBetweenSticks`, which contains counts of weeks between needle sticks that were computed from the `NeedleSticks` data. Because only five needle sticks were recorded, there are four counts of weeks between sticks.

**Figure 3** The Data Set `WeeksBetweenSticks`

<u>WeeksBetween</u>
.
5
14
0
24

The following statements create a rare events chart for the variable **WeeksBetween**:


```
proc rareevents data=WeeksBetweenSticks;
  chart WeeksBetween / odstitle='Rare Events Chart for Needle Sticks';
run;
```

Because the data are integers, the RAREEVENTS procedure fits a geometric distribution to the data and computes probability limits based on this distribution. The probability that a variable with this distribution is less than the lower probability limit (LPL) is  $\alpha_{LPL}$ , and the probability that a variable with this distribution is greater than the upper probability limit (UPL) is  $\alpha_{UPL}$ . By default,  $\alpha_{LPL} = \alpha_{UPL} = 0.005$ . Therefore, there is 1 chance in 100 of a measurement outside the limits; this implies unusual variation. You can specify smaller values for  $\alpha_{LPL}$  and  $\alpha_{UPL}$  to set a stricter standard for unusual variation, or you can specify larger values to set a looser standard.

A value that is greater than the UPL indicates that the time between events is increasing, in which case the events are occurring less frequently. Because the events of interest are usually adverse, this can signal an improvement in the process. Conversely, a value less than the LPL indicates that events are occurring more frequently, which can signal a deterioration in the process.

The resulting rare events chart is shown in Figure 4.

**Figure 4** Rare Events Chart of Weeks between Needle Sticks



Note that the LPL in Figure 4 is zero, which means that no value from the geometric distribution can be less than the LPL. When this is the case, the RAREEVENTS procedure checks for sequences of consecutive values equal to the LPL. A sequence whose probability of occurring is  $\alpha_{LPL}$  is flagged as an indication of unusual variation.

Because values from the geometric distribution are integers, it is not usually possible to compute probability limits that correspond exactly to the intended  $\alpha_{LPL}$  and  $\alpha_{UPL}$ . The limits are computed so that the associated probabilities are as close as possible to the intended values without exceeding them. The legend below the chart shows the actual  $\alpha_{LPL}$  and  $\alpha_{UPL}$  that correspond to the limits. The label in the upper right corner of the chart displays the total probability of unusual variation,  $\alpha = \alpha_{LPL} + \alpha_{UPL}$ .

The chart in Figure 4 indicates no unusual variation.

## MONITORING URINARY TRACT INFECTIONS

A hospital system tracked the frequency of urinary tract infections (UTIs) that were acquired by patients while in one of its hospitals (Santiago and Smith 2013). The following statements create a SAS data set that contains the variable

**DaysBetween**, which represents the number of days between discharges from the hospital of patients who acquired UTIs:


```
data UrinaryTractInfections;
  input DaysBetween @@;
  label DaysBetween = 'Days between UTIs';
datalines;
0.57014 0.07431 0.15278 0.14583 0.13889
0.14931 0.03333 0.08681 0.33681 0.03819
0.24653 0.29514 0.11944 0.05208 0.12500
0.25000 0.40069 0.02500 0.12014 0.11458
0.00347 0.12014 0.04861 0.02778 0.32639
0.64931 0.14931 0.01389 0.03819 0.46806
0.22222 0.29514 0.53472 0.15139 0.52569
0.07986 0.27083 0.04514 0.13542 0.08681
0.40347 0.12639 0.18403 0.70833 0.15625
0.24653 0.04514 0.01736 1.08889 0.05208
0.02778 0.03472 0.23611 0.35972
;
```

The following statements produce a graph that compares the data distribution to a theoretical distribution that is estimated from the data. In this example, the RAREEVENTS procedure uses an exponential distribution by default because these data are continuous.

```
proc rareevents data=UrinaryTractInfections;
  compare DaysBetween / nbins=12;
run;
```


The NBINS= option specifies that 12 histogram bins be used to display the data. [Figure 5](#) shows the resulting histogram of the data overlaid with the exponential curve.

**Figure 5** Distribution of Days between UTIs



The histogram indicates that an exponential distribution is appropriate for the data. It is good practice to verify that the assumed distribution reasonably matches the data distribution so that the rare events chart's probability limits are valid. Figure 6 shows a rare events chart for the UTI data.

**Figure 6** Rare Events Chart for Days between UTIs



Note that because the exponential distribution is continuous, the probability limits correspond exactly to the specified (or default)  $\alpha_{LPL}$  and  $\alpha_{UPL}$  values. The rare events chart shows no indication of unusual variation in the incidence of UTIs among these patients.

## MONITORING MORTALITY ASSOCIATED WITH A CONGENITAL HEART DEFECT

Hypoplastic left heart syndrome (HLHS) is a rare congenital condition in which the left side of the heart is critically underdeveloped. The typical treatment for this condition consists of a series of surgeries: one shortly after birth, another several months later, and a third after several years. During the time between the first two surgeries, which is referred to as the interstage period, there is a high risk of mortality. Because HLHS is so rare, evaluating the effectiveness of different treatment options in reducing interstage mortality is challenging.

The National Pediatric Cardiology Quality Improvement Collaborative (NPC-QIC) was created to improve the outcomes of patients with HLHS (Anderson et al. 2015). By collecting data from over 50 participating hospitals, NPC-QIC can learn more about treating this rare condition than an individual hospital could on its own. NPC-QIC researchers monitor various measures of patient outcomes by using various types of control charts, including  $g$  charts, which are similar to the rare events charts that the RAREEVENTS procedure produces.

Figure 7 shows a partial listing of the data set Interstage, which contains interstage mortality data that NPC-QIC collected from February 21, 2012, to July 9, 2015. Each observation records the date of an interstage mortality and the number of infants who survived their interstage period between that death and the previous one.

**Figure 7** Partial Listing of the Interstage Data Set


Date	PatientsBetween
21FEB2012	18
19MAR2012	19
18APR2012	21
22APR2012	2
03MAY2012	9
04MAY2012	0
16MAY2012	10
24MAY2012	6
03JUN2012	5
03JUN2012	0

The following statements produce a chart that compares the interstage data to a geometric distribution:

```
proc rareevents data=Interstage;  
  compare PatientsBetween;  
run;
```

Because the values of **PatientsBetween** are integers, the RAREEVENTS procedure produces a needle chart to compare the data to a geometric distribution. The chart is shown in [Figure 8](#). The estimated distribution represents the data reasonably well.

**Figure 8** Distribution of Patients Treated between Interstage Mortalities




You can use rare events charts to analyze the HLHS interstage data and to identify unusual variation in interstage mortality. A significant improvement in interstage survival would be identified by unusual variation above a chart's UPL. If you create a series of rare events charts, adding one observation from the Interstage data set for each successive chart, the first indication of unusual variation occurs at June 7, 2013. The following statements create a rare events chart of the interstage data up to that point:

```
proc rareevents data=Interstage;
  where Date <= '07jun2013'd;
  chart PatientsBetween * Date / totpanels=1;
run;
```

**Date** is specified as an optional index variable that identifies the data points. The resulting chart is shown in [Figure 9](#).

**Figure 9** Indication of Unusual Variation in Interstage Mortality



The point above the UPL indicates that changes in the treatment process might have reduced interstage mortality, but you need further analysis to verify that conclusion. The following statements produce a chart of data before June 7 and save the limits in the data set Limits1:

```
proc rareevents data=Interstage;
  where Date < '07jun2013'd;
  chart PatientsBetween * Date / totpanels=1 outlimits=Limits1;
run;
```


The following statements use the saved probability limits and produce a chart of data from June 7 and later:

```
proc rareevents data=Interstage limits=Limits1;
  where Date >= '07jun2013'd;
  chart PatientsBetween * Date / totpanels=1;
run;
```


The two charts are shown in [Figure 10](#) and [Figure 11](#).



**Figure 10** Interstage Mortality before June 7, 2013




**Figure 11** Interstage Mortality: June 7 and After



As described previously, the chart shown in [Figure 10](#) indicates no unusual variation. However, the chart shown in [Figure 11](#) tells a different story. Of the 37 data points in the chart, 3 points exceed the UPL, accounting for just over 8% of the total—more than 16 times the expected rate, based on  $\alpha_{UPL} = 0.0049$ . This is a strong indication that the data from before and after June 7 come from two different distributions. There has been a change in the process: more patients are surviving the interstage period.

Finally, [Figure 12](#) shows a rare events chart for the data beginning June 7, with probability limits computed from those data only.

**Figure 12** Results from the Improved Interstage Treatment Process



The rare events chart indicates that the new, improved process is stable. The center line of the chart marks the median of the estimated distribution. If you compare [Figure 10](#) to [Figure 12](#), you can see that the median increased from 7.04 to 13.59. The mean number of patients between mortalities increased from 9.3 to 18.6.

## CONCLUSION

The RAREEVENTS procedure in SAS/QC software creates rare events charts with probability limits based on geometric or exponential distributions. It also produces histograms and needle plots that are used to determine whether the data are adequately described by the distribution.

When you are analyzing counts of events, a rare events chart is a useful alternative to a traditional control chart for monitoring low-probability events. Instead of plotting counts of events, a rare events chart plots counts of “missed opportunities” or measures of elapsed time between events. If the counts in your samples show sufficient variation, then you should use a traditional control chart, which you can produce with the SHEWHART procedure. If you find that a traditional chart identifies the occurrence of a single event as unusual variation, you should consider using the RAREEVENTS procedure instead.

## REFERENCES

- Anderson, J. B., Beekman, R. H., Kugler, J. D., Rosenthal, G. L., Jenkins, K. J., Klitzner, T. S., Martin, G. R., Neish, S. R., Brown, D. W., Mangeot, C., King, E., Peterson, L. E., Provost, L., and Lannon, C. (2015). "Improvement in Interstage Survival in a National Pediatric Cardiology Learning Network." *Circulation: Cardiovascular Quality and Outcomes* 8:428–436.
- Benneyan, J. (2001). "Number-Between  $g$ -Type Statistical Control Charts." *Health Care Management Science* 4:305–318.
- Benneyan, J. C. (1999). "Geometric-Based  $g$ -Type Statistical Control Charts for Infrequent Adverse Events: New Quality Control Charts for Hospital Infections." In *Institute of Industrial Engineers Society for Health Systems 1999 Conference Proceedings*, 175–185. Norcross, GA: Institute of Industrial Engineers, Society for Health Systems.
- Kaminsky, F. C., Benneyan, J. C., Davis, R. D., and Burke, R. J. (1992). "Statistical Control Charts Based on a Geometric Distribution." *Journal of Quality Technology* 24:63–69.
- Santiago, E., and Smith, J. (2013). "Control Charts Based on the Exponential Distribution: Adapting Runs Rules for the  $t$  Chart." *Quality Engineering* 25:85–96.
- Wilburn, S. Q., and Eijkemans, G. (2004). "Preventing Needlestick Injuries among Healthcare Workers: A WHO-ICN Collaboration." *International Journal of Occupational and Environmental Health* 10:451–456.
- Woodall, W. H. (2006). "The Use of Control Charts in Health-Care and Public-Health Surveillance." *Journal of Quality Technology* 38:89–104.

## ACKNOWLEDGMENTS

The author is grateful to Colleen Mangeot of the Cincinnati Children's Hospital Medical Center, and Bob Rodriguez and Ed Huddleston of the Advanced Analytics Division at SAS Institute Inc. for their valuable assistance in the preparation of this paper.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Bucky Ransdell  
SAS Institute Inc.  
SAS Campus Drive  
Cary, NC 27513  
919-521-7928  
Bucky.Ransdell@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.