


Simple GTL Example

```
proc sort data=sashelp.class out=class;
  by sex;
run;

proc template;
  define statgraph groupreg;
    begingraph;
      entrytitle 'Separate Fit By Sex';
      layout overlay;
        scatterplot y=weight x=height /
          group=sex;
        regressionplot y=weight x=height /
          group=sex degree=3 name='reg';
        discretelegend 'reg' / title='Sex';
      endlayout;
    endgraph;
  end;
run;

proc sgrender data=class template=groupreg;
run;
```



Other GTL Statements


notes *“template-description”*;
 dynamic *dynamic-variable list*;
 mvar *macro-variable list*;
 nmvar *numeric-macro-variable list*;
 if (condition) *GTL-statements*
 else *GTL-statements*
 endif;
 sidebar / *<options>*; *GTL-statements* endsidebar;
 cell / *<options>*; *GTL-statements* endcell;

Layout Lattice Example

```
proc template;
  define statgraph res;
    begingraph;
      entrytitle 'Residuals by Predictor';
      layout lattice / rows=2 columns=2;
        layout overlay;
          scatterplot y=r x=height / group=sex;
          loessplot y=r x=height;
        endlayout;
        layout overlay;
          boxplot y=r x=sex;
        endlayout;
        layout overlay;
          scatterplot y=r x=age / group=sex;
          loessplot y=r x=age;
        endlayout;
      endlayout;
    endgraph;
  end;
run;

proc glm data=sashelp.class;
  class sex;
  model weight = height sex age;
  output residuals=r out=r;
run; quit;

proc sgrender data=r template=res;
  label r= 'Residuals';
run;
```



For more information, see:
 Kuhfeld, W. F. 2010. *Statistical Graphics in SAS[®]: An Introduction to the Graph Template Language and the Statistical Graphics Procedures*. Cary, NC: SAS Press.
support.sas.com/publishing/authors/kuhfeld.html

For complete information, see the SAS[®] 9.2 documentation at
<http://support.sas.com/v9doc>



Graph Template Language Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

ODS Graphics is an extension of ODS (the Output Delivery System). The Graph Template Language (GTL) specifies the layout and details of each graph produced by ODS. This powerful language includes statements for specifying plot layouts (such as lattices or overlays), plot types (such as scatter plots and histograms), and text elements (such as titles, footnotes, and insets). It also provides support for built-in computations (such as histogram binning) and the evaluation of expressions. Visual attributes of graphs are determined by the active ODS style. However, options are available for specifying colors, marker symbols, and other plot attributes.

This tip sheet presents the statements and options used that are most commonly used with the Graph Template Language in SAS[®] 9.2.



THE
POWER
TO KNOW[®]

SAS Institute Inc. World Headquarters
 +1 919 677 8000 www.sas.com/offices

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2011, SAS Institute Inc. All rights reserved. 295417_0211

SAS[®]9 Graph Template Language Tip Sheet

GTL Overview

Each graph has an ODS template—a SAS program that provides instructions for creating the graph. SAS provides a template for every graph. You do not need to know *anything* about templates to create graphs. However, you can easily modify templates and make permanent changes that apply every time that you run SAS.

Graph Template Structure

```
proc template;
  define statgraph template-name;
    begingraph / <designheight=h>
      designwidth=w> <options>;
      GTL-global-statements
      GTL-layout-block
    endgraph;
  end;
run;
```

Legend Statements

```
continuouslegend names / <options>;
discretelegend names / <options>;
```

Text Statements

```
entry text-items / <options>;
entryfootnote text-items / <options>;
entrytitle text-items / <options>;
```

Linear Axis Options

```
integer=boolean
origin=number
thresholdmax=number
thresholdmin=number
tickvaluefitpolicy=thin | other-policy
tickvalueformat=(format-options) | data | format
tickvaluelist=(numeric-list)
tickvaluepriority=boolean
tickvaluesequence=(start=s end=e increment=i)
viewmax=number
viewmin=number
```

Layout Statements

```
layout dataattice rowvar=r-var columnvar=c-var
  / <options>;
layout datapanel classvars = (c-vars) / <options>;
layout gridded / <options>;
layout lattice / <columns=nc> <rows=nr>
  <rowdatarange=data | union | unionall>
  <columndatarange=data | union | unionall>
  <rowweights=(list)> <columnweights=(list)>
  <rowgutter=rg> <columngutter=cg> <options>;
layout overlay / <xaxisopts=(axis-options)>
  <yaxisopts=(axis-options)>
  <x2axisopts=(axis-options)>
  <y2axisopts=(axis-options)> <options>;
layout overlayequated /
  <equatetype=square | fit | equate>
  <commonaxisopts=(axis-options)>
  <xaxisopts=(axis-options)>
  <yaxisopts=(axis-options)> <options>;
layout overlay3d / <options>;
layout prototype / <options>;
```

Axis Options

```
discreteopts=(discrete-axis-options)
display=standard | all | none | (display-options)
displaysecondary=none | all | standard |
  (display-options)
gridattrs=style-element (line-options)
griddisplay=auto_off | auto_on | on | off
label="string"
labelattrs=<style-element> (text-options)
linearopts=(linear-axis-options)
logopts=(log-axis-options)
offsetmax=auto | autocompress | number
offsetmin=auto | autocompress | number
reverse=boolean
shortlabel="string"
tickstyle=outside | inside | across
timeopts=(time-axis-options)
type=auto | discrete | linear | time | log
```

Plot Statements

```
bandplot y|x=var limitupper=u-var
  limitlower=l-var /
  <outlineattrs=<style-element>(line-options)>
  <options>;
barchart x=x-var / <stat=stat> <group=g-var>
  <orient=vertical | horizontal>
  <outlineattrs=<style-element>(line-options)>
  <options>;
barchartparm y=y-var x=x-var / <group=g-var>
  <orient=vertical | horizontal>
  <outlineattrs=<style-element>(line-options)>
  <options>;
bihistogram3dparm y=y-var x=x-var z=z-vars /
  <options>;
blockplot x=x-var block=b-var / <options>;
boxplot y|x=var / <orient=vertical | horizontal>
  <options>;
boxplotparm y|x=var stat=s-var /
  <orient=vertical | horizontal> <options>;
contourplotparm y=y-var x=x-var z=z-vars /
  <options>;
densityplot var / <kernel(options)>
  <normal(options)> <options>;
dropline y=y-var x=x-var / dropto=x|y
  <label=l-var> <options>;
ellipse y=y-var x=x-var / <options>;
ellipseparm xorigin=x yorigin=y
  semimajor=s1-var semiminor=s2-var
  slope=slope / <options>;
fringeplot var / <options>;
histogram var / <orient=vertical | horizontal>
  <options>;
histogramparm y=y-var x=x-var /
  <orient=vertical | horizontal> <options>;
lineparm y=y-var x=x-var slope=s /
  <group=g-var> <options>;
loessplot y=y-var x=x-var / <alpha=a>
  <clm="name"> <degree=1|2> <smooth=auto|s>
  <interpolation=linear|cubic> <maxpoints=m>
  <lineattrs=<style-element>(line-options)>
  <group=g-var> <options>;
```

Plot Statements

```
modelband name / <options>;
needleplot y=y-var x=x-var / <group=g-var>
  <options>;
pbsplineplot y=y-var x=x-var / <alpha=a>
  <cli="name"> <clm="name"> <degree=d>
  <freq=f-var> <maxpoints=m> <nknots=n>
  <smooth=auto|s> <weight=w-var>
  <lineattrs=<style-element>(line-options)>
  <group=g-var> <options>;
referenceline x|y=var / <options>
  <lineattrs=<style-element>(line-options)>;
regressionplot y=y-var x=x-var / <alpha=a>
  <cli="name"> <clm="name"> <degree=d>
  <freq=f-var> <maxpoints=m> <weight=w-var>
  <lineattrs=<style-element>(line-options)>
  <group=g-var> <options>;
scatterplot y=y-var x=x-var / <datalabel=l-var>
  <group=g-var> <markercharacter=m-var>
  <markerattrs=<style-element>(line-options)>
  <options>;
scatterplotmatrix var / <datalabel=l-var>
  <group=g-var> <markercharacter=m-var>
  <diagonal=(<histogram> <normal> <kernel>)>
  <markerattrs=<style-element>(line-options)>
  <options>;
seriesplot y=y-var x=x-var / <group=g-var>
  <display=standard | all>
  <lineattrs=<style-element>(line-options)>
  <markerattrs=<style-element>(line-options)>
  <options>;
stepplot y=y-var x=x-var / <datalabel=l-var>
  <lineattrs=<style-element>(line-options)>
  <markerattrs=<style-element>(line-options)>
  <group=g-var> <options>;
surfaceplotparm y=y-var x=x-var z=z-vars /
  <options>;
vectorplot y=y-var x=x-var xorigin=0 yorigin=0 /
  <datalabel=l-var> <group=g-var>
  <lineattrs=<style-element>(line-options)>
  <options>;
```